/*
 * Copyright 2016 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

package org.gradle.integtests.tooling.r32

import org.gradle.integtests.tooling.fixture.TargetGradleVersion
import org.gradle.integtests.tooling.fixture.ToolingApiSpecification
import org.gradle.tooling.BuildAction
import org.gradle.tooling.BuildController
import org.gradle.tooling.model.GradleProject

@TargetGradleVersion(">=3.2")
class BuildFinishedCrossVersionSpec extends ToolingApiSpecification {

    def "model builders are executed before buildFinished hook"() {
        when:
        def project = loadToolingModel(GradleProject)

        then:
        fetchedBeforeBuildFinished(project)
    }

    def "build actions are executed before buildFinished hook"() {
        when:
        GradleProject project = withConnection {
            action(new GetGradleProject()).run()
        }

        then:
        fetchedBeforeBuildFinished(project)
    }

    void setup() {
        buildFile << """
            gradle.buildFinished {
                buildDir = "someOtherDir"
            }
        """
    }

    private boolean fetchedBeforeBuildFinished(GradleProject project) {
        project.buildDirectory == file("build")
    }

    private static class GetGradleProject implements BuildAction<GradleProject> {

        @Override
        GradleProject execute(BuildController controller) {
            controller.getModel(GradleProject)
        }
    }
}
